/*
 * Copyright (c) 2008-2010, JFXtras Group
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 * 1. Redistributions of source code must retain the above copyright notice,
 *    this list of conditions and the following disclaimer.
 * 2. Redistributions in binary form must reproduce the above copyright notice,
 *    this list of conditions and the following disclaimer in the documentation
 *    and/or other materials provided with the distribution.
 * 3. Neither the name of JFXtras nor the names of its contributors may be used
 *    to endorse or promote products derived from this software without
 *    specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
 * AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
 * IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
 * ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT OWNER OR CONTRIBUTORS BE
 * LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR
 * CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF
 * SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS
 * INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN
 * CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
 * ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE
 * POSSIBILITY OF SUCH DAMAGE.
 */
package org.jfxtras.scene.control.renderer;

import javafx.scene.Node;
import javafx.scene.image.Image;
import javafx.scene.image.ImageView;
import org.jfxtras.lang.XBind;
import org.jfxtras.scene.image.XImage;
import org.jfxtras.util.XMap;

/**
 * @author Stephen Chin
 */
// todo - need to limit the size of the image cache, also should it be exposed?
package var imageCache:XMap = XMap {}
public class ImageRenderer extends CachingNodeRenderer {
    public var placeholder:Image;
    public var missing:Image;

    override function createNode():Node {
        return ImageView {}
    }

    override function populateNode(node:Node, data:XBind, index:Integer, width:Number, height:Number):Void {
        def imageView = node as ImageView;
        def url = data.ref as String;
        def cached = if (url == "") missing else imageCache.get(url) as Image;
        imageView.fitWidth = width;
        imageView.fitHeight = height;
        imageView.preserveRatio = true;
        imageView.image = if (cached != null) cached else XImage {
            backgroundLoading: true
            preserveRatio: true
            url: url
            width: width
            height: height
            placeholder: placeholder
        }
    }

    override function recycleNode(node:Node) {
        def imageView = node as ImageView;
        if (imageView.image != missing and not imageCache.containsKey(imageView.image.url)) {
            imageCache.put(imageView.image.url, imageView.image);
        }
        super.recycleNode(node);
    }
}
